
1 

SESUG 2021 Paper 054 

Identifying Data Inconsistencies with SAS® 
Imelda C. Go, Cognia, Inc. 

 

ABSTRACT  
This paper goes over a quality control example on how to identify data inconsistencies when 
the expectation is data consistency. PROC MEANS is used to diagnose data inconsistency 
and to generate an output data set. This data set is then presented to the reader using 
PROC TABULATE, which helps present the inconsistencies in a way different from viewing a 
data set. For example, a test item appears on different test forms. The metadata for the 
same item are expected to be identical across test forms where the item appears. How can 
you easily confirm that the item metadata are identical across all test forms and how can 
you identify inconsistencies in a report that facilitates the identification and correction of the 
inconsistencies? With macro techniques, we can further automate this process to include 
output via ODS Excel for archiving and sharing our findings. 

INTRODUCTION  
The method uses PROC MEANS predominantly to identify the inconsistencies. An example 
from the assessment industry will be used to illustrate the technique. Let us assume the 
following: 

 We have 3 mathematics test forms (A-C) with 3 items each (2 operational (OP) test 
items and 1 field (FT) test). 

 The 1st and 3rd items are both OP items, while the 2nd item is an FT item. 
 The expectation is that the same item (based on ItemID), wherever it appears 

among the forms, has the same AnswerKey, ItemFunction, ItemType, and Position 
values. 
 

Subject FormID ItemID AnswerKey ItemFunction ItemType Position
math A 11 A OP MC 1 
math A 33 Yellow FT CR 2 
math A 22 C OP MC 3 
math B 11 A OP CR 1 
math B 44 Red FT CR 2 
math B 22 C OP MC 3 
math C 11 A OP CR 1 
math C 55 Blue FT CR 3 
math C 22 B OP MC 2 

 

Given the above data, we see the following issues: 

 For ItemID=22, the AnswerKey is C on two forms and is B on form C.  
 For ItemID=11, the ItemType value is inconsistent. It is MC on one form and CR on 

the other two. 
 ItemID=11 and ItemID=22 appear as the 1st and 3rd items on forms A and B but not 

on form C. 

For such a short test, we can examine the data visually with ease. We want a solution, 
which tells us exactly where the inconsistencies are quickly and with less effort.  
 
For the sake of providing an example, let us also expand the scenario to using these items 
for a 3rd and 4th grade test. (The 3rd and 4th grade test have the exact same items.) Also 
assume that there are no missing values for each variable. The sample code was prepared 
with the intent to illustrate the technique. In actual practice, certain parts of it can be 
rewritten more succinctly (e.g., PROC SQL coding) or be adjusted (e.g., missing values are 
present). 
 
We will process the 3rd and 4th grade test data set using PROC MEANS. Without a VAR 
statement, PROC MEANS will attempt to use all numeric variables on the data set. To make 
things simpler, add a numeric variable called NumRecords with a value of 1 to the data set.  
 

 
 
We will use a combination of PROC MEANS, PROC TABULATE, and ODS Excel to create 
output, regarding data inconsistencies, and send the output to an Excel file with multiple 
worksheets. 

The only input required from the user is shown below in the six %LET statements. Beyond 
that, the code is data-driven/generalized and will function based on the values in the %LET 
statements. 
 
%let InputData=Sample; 
 
%let DataVars=AnswerKey ItemFunction 
ItemType Position; 
 
%let IDVar=ItemID; 
 
%let FormVar=FormID; 
 
%let GroupVars=Subject &IDVar.; 
 
%let OutPath=C:\Users\imelda.go\Desktop; 
 
 
 
 

&InputData specifies the data set. 
 
&DataVars is a list of variables that 
need to be checked for consistency.   
 
&IDVar is the item ID.  
 
&FormVar is the test form ID. 
 
&GroupVars lists the groups of items 
where you want to check 
inconsistencies for. It is at minimum 
the &IDvar.  
 
By specifying Subject prior to &IDVar, 
you will be checking for inconsistencies 
within each value of the subject 
variable. Leave &IDVar in the last 
position in the list since &GroupVars 
will be used in a CLASS statement 
where the order of variables listed 
affects the _type_ variable generated 
by PROC MEANS. 

%let GroupVarsAsterisk =  
%sysfunc(translate(&GroupVars,'*',' ')); 
 
%let GroupVarsComma = 
%sysfunc(translate(&GroupVars,',',' ')); 
 
%let 
NDataVars=%sysfunc(countw(&DataVars)); 
 
%let 
NGroupVars=%sysfunc(countw(&GroupVars)); 
  

&GroupVarsAsterisk is the 
&GroupVars value with an asterisk (*) 
as the delimiter, which will be used in 
TYPES statement. 
 
&GroupVarsComma is the &GroupVars 
value with a comma (,) as the 
delimiter, which will be used in PROC 
SQL. 
 
&NDataVars is the number of variables 
listed in &DataVars, which will be used 
for automation (DO loop statement). 
 
&NGroupVars is the number of 
variables in listed in &GroupVars, which 
will be used to calculate the value of 
_type_ that we need to identify 
records of interest.  

data TypeValue; 
retain type 0; 
if &NGroupVars>0 then 
do i=1 to &NGroupVars; 
type=type+ 2**(i-1); 
end; 
output; 
call symputx('type',type); 
 

Calculate the value of _type_ that 
corresponds to the record with the 
total number of times a combination 
of values appear for variables listed in 
&GroupVars. 
 
In this example &GroupVar=subject 
ItemID so we are counting the 
number of records per subject and 
itemID combination. In this case the 
value of interest is _type_=3. It is 3 
because we have 2 variables in 
&GroupVars. Since these are the last 
two variables listed in the CLASS 
statement, the corresponding _type_ 
value will be 21+20=2+1=3.  

proc means data=&InputData noprint n 
chartype; 
id &FormVar; 
class &DataVars &GroupVars/missing; 
types &GroupVarsasterisk. 
&GroupVarsasterisk.*(&DataVars); 
var NumRecords; 
output out=CountCombos (where=( 
_stat_="N" )); 
 
 

PROC MEANS allows us to count 
various things. We can limit what 
we count by using the TYPES 
statement.  
 
The ID statement adds the 
&FormVar variable to the output 
data set. 
 
The order in which the variables 
are listed in the CLASS statement 
affect the values of the _type_ 
variable. Due to the need for 
predictability, put &ItemGroupVars 
at the very end of the list of 
variables in the CLASS statement. 
You need to know the right _type_ 
value to use in the next step. 
 
The data were sorted for 
instructional purposes to illustrate 
below the rationale for the 
technique.  

The sorted data set is shown below.  
 

 Whenever _type_=3, NumRecords is equal to the total number of records that appear 
for each unique combination of subject and ItemID. In general, the data set tells us 
the number of unique combinations listed in the CLASS statement but specifically the 
combinations of interest specified in the TYPES statement. The _type_ value tells us 
exactly which combination of variables is involved for each row in the data set. The 
specific values for each combination of interest are in the data set. 

 
 The variables in the CLASS statement are AnswerKey, ItemFunction, ItemType, 

Position, Subject, ItemID. The sample data are such that NumRecords is always 
6 if the data are all consistent. Wherever we see NumRecords not equal to 6, we see 
inconsistencies in the values as shown. Where there are inconsistencies, note that 
the _type_ values occur more than once among variable combinations.  If the 
_type_ value occurs only once among the combinations of those variables, then the 
data were consistent. We can also see which test forms these inconsistencies occur 
because we used the ID statement. 
 

  

The value of _type_ depends on the order in which the variables in the CLASS statement 
are listed. The decimal value of _type_ appears as the default and using the PROC MEANS 
CHARTYPE option will produce a character string representative of binary notation, which is 
equivalent to the decimal value of _type_. For this reason, the &GroupVars was placed at 
the end of the CLASS statement to increase predictability of the value and provide 
automation opportunities. 
 

AnswerKey ItemFunction ItemType Position Subject ItemID 

Decimal 
_type_ 
without 
chartypes 
PROC 
MEANS 
option 

Binary
_type_ 
with 
chartypes 
PROC 
MEANS 
option 

0 0 0 0 1 1 3 000011 
0 0 0 1 1 1 7 000111 
0 0 1 0 1 1 11 001011 
0 1 0 0 1 1 19 010011 
1 0 0 0 1 1 35 100011 

 
%macro filter; 
proc sort data=CountCombos; by &GroupVars 
_type_; 
 
data CheckThese ; 
length Flag $30.; 
retain TotalRecords . ; 
set CountCombos; 
by &GroupVars _type_; 
if first.&IDVar and _type_=&type then 
do; TotalRecords=NumRecords; delete; end; 
 

This macro will create a flag 
variable that will specify the 
variable with inconsistency. 
 
When _type_=3, then NumRecords is 
the TotalRecords for each subject 
and ItemID combination. 
 
When _type_ is not 3, then it is the 
record that counts the unique 
combination of values for the 
corresponding variables. 

if NumRecords ne TotalRecords then do; 
  %do i = 1 %to &NDataVars; 
   %let column=%scan(&DataVars,&i); 
    if &column ne "" then  
Flag=strip(Flag)||"+&column"; 
   %end; 
output; drop _stat_ _type_; end; 
run; 
%mend filter; 
 
%filter; 
 

Check if the record needs to be 
flagged. That is, check if 
NumRecords is not equal to 
TotalRecords for the corresponding 
&ItemVar variable. Only keep 
flagged records. If 
Numrecords=TotalRecords, that tells 
us that there was only one value 
for the column (i.e., data value 
was consistent for that 
combination of subject and 
ItemID). 
 
If it is a flagged record, the flag 
value will contain the name of the 
variable with inconsistencies.  

Dataset checkthese has the following contents.  
 

 
 
We will now take the checkthese data and do a few things to it to produce another way of 
looking at the data.  
 
proc sql; 
create table UniqueIDs 
as select unique Subject, &IDVar 
,TotalRecords from CheckThese 
where NumRecords ne TotalRecords; 
 

This is the list of items with flagged 
inconsistencies. 

 

 
 
proc sql; 
create table PreFinal as 
select a.TotalRecords, b.* 
from UniqueIDs as a left join &InputData 
(drop=NumRecords) as b 
on a.&IDVar=b.&IDVar and 
a.Subject=b.Subject; 
 

Add the item data to this list of unique 
IDs with inconsistencies. 

proc sql; 
create table UniqueFlags 
as select unique Subject, ItemID , Flag 
from CheckThese; 
 

Get all the unique subject, ItemID, 
and flag combinations from 
checkthese dataset. 

 

 
 

proc sql; 
create table Final as 
select a.Flag, b.* 
from UniqueFlags as a LEFT JOIN PreFinal 
as b 
ON a.&IDVar=b.&IDVar and 
a.Subject=b.Subject; 
quit; 
 

This is the data with all the original 
item data with the flagged records 
together with the flag values. The 
flag values tell us which variable is 
being flagged for each row. 

 
In this data set, we can see all the item data together with the flag value that tells us what 
the inconsistency is. In the first six records, the flag indicates there is an inconsistency in 
ItemType values for itemID. In the We see that the ItemType is MC in FormID=A and is CR 
on the other forms. 

%macro GenerateTables(column); 
proc tabulate data=Final out=test 
format=12.0; 
class Flag &GroupVars &FormVar &DataVars 
TotalRecords; 
table  
&GroupVarsAsterisk.*(&column)*TotalRecor
ds="# of &InputData Records with 
ItemID", all='Number of Records with 
Combination' FormID; 
where find(Flag,"&column")>0; 
title4 "&column: Item Inconsistencies 
Across Forms"; 
run; 
 
proc tabulate data=&InputData 
format=12.0; 
class &GroupVars &FormVar &DataVars ; 
table  &GroupVarsasterisk.*(&column), 
all FormID; 
title4 "&column: Data 
Consistency/Inconcistency Across Forms"; 
run; 
%mend GenerateTables; 
 

For each variable specified, up to two 
PROC TABULATE output tables are 
generated.  

 

The first PROC TABULATE output shows 
us the inconsistencies only as they 
appear in the final data set. This uses 
a WHERE statement and therefore, 
may not produce any output if no 
records satisfy the condition. 

 

The second one serves as a record for 
what is in the data and uses the 
original &dataset. Consistencies 
and/or inconsistencies will appear in 
the output.  

%macro feedback(varlist); 
  %do i = 1 %to &NDataVars; 
   %let column=%scan(&varlist,&i); 
    ods excel options 
(sheet_name="&column"); 
   %GenerateTables(column=&column); 
  %end; 
%mend feedback; 
 

The DO loop goes over each variable in 
&ndatavars and runs the 
%generatetables macro for each 
variable. 

ODS Excel options specified the 
worksheet name. 

Each DO loop iteration invokes the 
%generatetables macro for each 
variable in &varlist. 

ods excel file="&OutPath.\&groupvars 
feedback.xlsx" options 
(sheet_interval='table' 
embedded_titles='yes'); 
%feedback(varlist=&DataVars); 
ods excel close; 
 

We are going to send the output to an 
Excel file using ODS. 

We invoke the %feedback macro by 
defaulting the varlist to &DataVars, 
which is all the variables we wanted to 
check the consistency of.  

The macro variable &varlist has 4 variables in this example. There will be at most 4 x 2 of 
PROC TABULATE output tables. The first PROC TABULATE output table will appear as long as 
there were inconsistencies found for the particular variable. The second PROC TABULATE 
output will always appear since it uses the original data set will all values in it. The PROC 
TABULATE presentation of the information offers another way of looking at the data. 

 
We can see easily below that the itemID=22 row has two keys. The key of B appears 
in form C and the key of C appears in forms A and B.  
 

 
 
The following is similar to the above table but also contains the information for all itemID 
values. 

No inconsistencies were detected for ItemFunction. Only the second PROC TABULATE 
output was produced. 

 

 
 
The following table shows that there was an inconsistency in ItemType when ItemID=11. 

The following contains the information for ItemType consistency/inconsistency. 

 

 
 

Position also has inconsistencies for ItemID=22. 

The following shows Position values for all items. 

 

 
 

All of these output tables were sent to an Excel file with several worksheets, one for each PROC 
TABULATE output table. The worksheets for produced are as follows. 

The &GroupVar value helps organize the tables according to relevant pools of items. In this 
example, &GroupVar=Subject. This means we want to look at all the duplicates within each 
subject value. 
 

 
 
If we use &GroupVar=grade, the output will automatically adjust. Here we are looking at duplicates 

within each grade value. 
 

 
 

If we use &GroupVar=subject grade, we will be looking at duplicates within each 
combination of subject and grade. 

 

 
 



15 

When &GroupVar=&ItemVar, then we will be strictly looking at item inconsistencies by 
ItemID in this example. 
 

 
Here’s the SAS coding described above in one continuous section. 
 
%let InputData=Sample; 
%let DataVars=AnswerKey ItemFunction ItemType Position; 
%let IDVar=ItemID; 
%let FormVar=FormID; 
%let GroupVars=Subject &IDVar.; 
%let OutPath=C:\Users\imelda.go\Desktop; 
***********************************; 
%let GroupVarsAsterisk = %sysfunc(translate(&GroupVars,'*',' ')); 
%let GroupVarsComma = %sysfunc(translate(&GroupVars,',',' ')); 
%let NDataVars=%sysfunc(countw(&DataVars)); 
%let NGroupVars=%sysfunc(countw(&GroupVars)); 
 
data TypeValue; 
retain type 0; 
if &NGroupVars>0 then 
do i=1 to &NGroupVars; 
type=type+ 2**(i-1); 
end; 
output; 
call symputx('type',type); 
 
proc means data=&InputData noprint n chartype; 
class &DataVars &GroupVars/missing; 
types &GroupVarsasterisk. &GroupVarsasterisk.*(&DataVars); 
var NumRecords; 
output out=CountCombos (where=( _stat_="N" )); 
 
proc sort data=CountCombos; by &GroupVars _type_;  

%macro filter; 
proc sort data=CountCombos; by &GroupVars _type_; 
 
data CheckThese ; 
length Flag $30.; 
retain TotalRecords . ; 
set CountCombos; 
by &GroupVars _type_; 
if first.&IDVar and _type_=&type then 
do; TotalRecords=NumRecords; delete; end; 
 
if NumRecords ne TotalRecords then do; 
  %do i = 1 %to &NDataVars; 
   %let column=%scan(&DataVars,&i); 
    if &column ne "" then  Flag=strip(Flag)||"+&column"; 
   %end; 
output; drop _stat_ _type_; end; 
run; 
%mend filter; 
 
%filter; 
 
proc sql; 
create table UniqueIDs 
as select unique Subject, &IDVar ,TotalRecords from CheckThese 
where NumRecords ne TotalRecords; 
 
proc sql; 
create table PreFinal as 
select a.TotalRecords, b.* 
from UniqueIDs as a left join &InputData (drop=NumRecords) as b 
on a.&IDVar=b.&IDVar and a.Subject=b.Subject; 
 
proc sql; 
create table UniqueFlags 
as select unique Subject, ItemID , Flag from CheckThese; 
 
proc sql; 
create table Final as 
select a.Flag, b.* 
from UniqueFlags as a LEFT JOIN PreFinal as b 
ON a.&IDVar=b.&IDVar and a.Subject=b.Subject; 
quit; 
 
%macro GenerateTables(column); 
proc tabulate data=Final out=test format=12.0; 
class Flag &GroupVars &FormVar &DataVars TotalRecords; 
table  &GroupVarsAsterisk.*(&column)*TotalRecords="# of &InputData Records 
with ItemID", all='Number of Records with Combination' FormID; 
where find(Flag,"&column")>0; 
title4 "&column: Item Inconsistencies Across Forms"; 
 
proc tabulate data=&InputData format=12.0; 
class &GroupVars &FormVar &DataVars ; 
table  &GroupVarsasterisk.*(&column), all FormID; 
title4 "&column: Data Consistency/Inconcistency Across Forms"; 
run; 
%mend GenerateTables; 

%macro feedback(varlist); 
  %do i = 1 %to &NDataVars; 
   %let column=%scan(&varlist,&i); 
    ods excel options (sheet_name="&column"); 
   %GenerateTables(column=&column); 
  %end; 
%mend feedback; 
 
ods excel file="&OutPath.\&groupvars feedback.xlsx" options 
(sheet_interval='table' embedded_titles='yes'); 
%feedback(varlist=&DataVars); 
ods excel close; 
 

CONCLUSION 
The SAS programming language offers different tools to identify data inconsistencies. The 
use of different tools together has the potential to result in greater automated and data-
driven coding efficiency.  

REFERENCES 
 
Lafler, Kirk Paul (2017). “Removing Duplicates Using SAS®”, Proceedings of the 2017 SAS 
Global Forum (SGF) Conference. 

CONTACT INFORMATION 
Your comments and questions are valued and encouraged. Contact the author at: 

Imelda C. Go, Ph.D.  
Cognia, Inc.  
imelda.go@cognia.org  
 

SAS and all other SAS Institute Inc. product or service names are registered trademarks or 
trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA 
registration. Other brand and product names are trademarks of their respective companies.  


