
Introduction

This is the overview, which consists of a list of topics in this section.overview

• what’s inside configuration files
• startup process
• why use configuration files

Configuration files contain two types of statements:what’s inside
allocation of environment variables and assignment of options.
The main configuration file contains over 200 lines; more than half of those
lines contain references to the environment variable sasroot in environ-
ment variables’ and options’ assignments.

startup process The startup process of SAS software has two insertion points for startup-
only options: on the command line, and in configuration files.
Table 1 lists the insertion points and the startup-only options.

Table 1 startup process
1. command-line: sas -sysin job-name <-options>
2. configuration files:

(a) primary: !sasroot/sasv9.cfg

(b) main: required !sasroot/nls/??/sasv9.cfg

(c) tertiary: project: sasv9.cfg

3. options from command-line

code-insertion points: startup-only options
option name default where set assignment
config sasv9.cfg -config 'filename.ext'
autoexec autoexec.sas -autoexec 'filename.ext'
initstmt command-line or config -initstmt '*text;'
sysparm command line: -sysparm "text"

sysin command line: -sysin job-name

sysparm macro variable, in program %let sysparm=value;

termstmt command-line or config -termstmt '*text;'

Description of this process is found in SAS Documentation under the topic:
Files Used by SAS.

Your installation may have several folders under National Language Sup-
port (nls); English (en) is shown here. See also the option locale.

why use config? Table 2 shows that almost one third of options are startup-only.

Table 2 frequency of optstart
Option Set

Cumulative Cumulative

optstart Frequency Percent Frequency Percent

-------- --------- ------- ---------- ----------

anytime 315 67.74 315 67.74

startup 150 32.26 465 100.00

Program 2, proc freq, tables optstart, pg. 5 produces this frequency table.

Bricolage: tools for discovery

values of opstart

There are two values of opstart: anytime and startup.overview
In the output of program proc-options-define-value-config.sas these values
are seen in the output table in the log in the section When Can Set:

When Can Set: Startup or anytime during the SAS Session

When Can Set: Session startup (command line or config) only

proc-options-define-
value-config.sas

*name: proc-options-define-value-config.sas;

proc options define value option=config;

run;

see line 23, for optstart='startup'log:

1 Option Value Information For SAS Option CONFIG

2 Value: ( "C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg"

3 "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"

4 "C:\SAS-projects\2021-book\sas-startup-configuration\sasv9.cfg"

5 )

6 Scope: SAS Session

7 How option value set: Internal

8

9 Option Definition Information for SAS Option CONFIG

10 Group= INSTALL

11 Group Description: Site installation settings

12 Description: Specifies the configuration file that is used

13 when initializing or overriding the values of SAS system options.

14 Type: The option value is of type CHARACTER

15 Maximum Number of Characters: 32000

16 Casing: The option value is retained with original casing

17 Quotes: If present during "set", start and end quotes are removed

18 Parentheses: The option value does not require enclosure

19 within parentheses.

20 If present, the parentheses are retained.

21 Expansion: Environment variables, within the option value,

22 are not expanded

23 When Can Set: Session startup (command line or config) only

echoing configuration files

This is the overview, which consists of a list of programs in this section.overview

• config syntax
• echo any file
• echo config primary
• echo config main

While viewing any configuration file, *.cfg, look for these two statements:config syntax

• allocate environment variable: -set evar value

• assign option: -<option-name> value

echo-any-file *name: echo-any-file.sas;
%put echo &=in_file;
%let lrecl = 80;
DATA _null_;

infile "&in_file" lrecl = &lrecl pad
end = endofile;

do until(endofile);
input @1 line $char&lrecl..;
put @1 line $char&lrecl..;
end;

stop;
run;

echo-config-primary
*name: echo-config-primary.sas;

%let in_file = %sysfunc(sysget(sasroot))\sasv9.cfg;

%include 'echo-any-file.sas';

log: echo IN_FILE=C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg

NOTE: The infile "C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg" is:
Filename=C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg,

-config "C:\program-files\SASHome\SASFoundation\9.?\nls\en\sasv9.cfg"

echo-config-main Program 1 echo-config-main.sas
*name: echo-config-main.sas;

%let in_file = %sysfunc(sysget(sascfg))\sasv9.cfg;

%include 'echo-any-file.sas';

log:
echo IN_FILE= C:\program-files\SASHome\SASFoundation\9.?\nls\en\sasv9.cfg
...
-SET SASROOT "C:\program-files\SASHome\SASFoundation\9.?"
...
-SET SASAUTOS ("!SASROOT\core\sasmacro"
...

"!SASROOT\graph\sasmacro"
"!SASROOT\stat\sasmacro"
)

The environment variable sasroot is used in the
environment variable sasautos, which is the list of folders of the

fileref sasautos, which is the default value of
option sasautos.

notes:

Note that lists of values, e.g. sasautos, are enclosed in parentheses.

sql procedure, dictionary.options

This is the overview, which consists of a list of topics in this section.overview

• describe dictionary options
• freq opstart

OnLine Doc: Customizing Your SAS Session by Using Configuration and
Autoexec Files

describe dictionary
options

*name: proc-sql-describe-dictionary-options.sas;

proc sql; describe table dictionary.options;

quit;

log: create table dictionary.options
optname char(32) label='Option Name',
opttype char(8) label='Option type',
offset num label='Offset into option value',
setting char(1024) label='Option Setting',
optdesc char(160) label='Option Description',
level char(8) label='Option Location',
optstart char(8) label='Option Set',
group char(32) label='Option Group'

Program 2 was used to produce table 2, shown above on pg. 2.freq optstart

Program 2 proc freq, tables optstart
*name: proc-freq-d-options-opstart.sas;

PROC sql; create table dictionary_options_optstart as

select optstart

from dictionary.options;

quit;

PROC freq data = &syslast;

tables optstart;

run;

Setup for testing
overview

This is the overview, which consists of a list of topics in this section.
This section shows how to create a suite of files for a unit test of a user-
written configuration file.

• functions used here
• write test suite, config
• sas.cmd
• sasv9.cfg
• autoexec.sas
• initstmt.sas
• termstmt.sas
• 01-config-test.*
• 02-autoexec-test.*

These functions are used in the following programs.functions used here
• getoption(option-name<,options>) fetch value of option
• putn(value,format) echo computed value
• %scan(string,n,delimiter) fetch word of string
• %sysfunc(fn(...)) access to data-step functions
• %sysget(evar) fetch value of environment variable (evar)

Program 3 writes to the specified folder a set of files for this article.write-test-suite-
config.sas

Program 3 write test suite of config test files
*name: write-test-suite-config.sas;

%let folder=c:\temp\sas-test;

%let folder=..\sas-startup-config-test-suite;

DATA _null_;

file "&folder\sas.cmd";

put 'rem name: sas.cmd '

/ '"' "%sysget(sasroot)\sas.exe" '" %*';

file "&folder\01-config-test.bat";

put 'rem name: 01-config-test.bat'

/ 'sas 01-config-test -verbose -pagesize max';

file "&folder\01-config-test.sas";

put '*name: 01-config-test.sas;';

file "&folder\sasv9.cfg";

put '/*name: sasv9.cfg */'

/ "-sasinitialfolder '.'"

/ "-set site_inc 'C:\SAS-projects\SAS-site\sas-includes' "

/ "-set site_inc '.'"

/ '-initstmt "' "%include '!site_inc\initstmt.sas';" '"'

/ '-termstmt "' "%include '!site_inc\termstmt.sas';" '"';

file "&folder\02-autoexec-test.bat";

put 'rem name: 02-autoexec-test.bat'

/ 'sas 02-autoexec-test -echoauto -source -source2';

file "&folder\02-autoexec-test.sas";

put '*name: 02-autoexec-test.sas;'

/ 'options pagesize=max;'

/ 'filename _all_ list;'

/ 'libname _all_ list;'

/ '%put echo mautosource=%sysfunc(getoption(mautosource));'

/ '%put %sysfunc(getoption(sasautos,keyword));'

/ '%let mvar2=text.2 for show sort order;'

/ '%let mvar1=text.1 for termstmt;';

file "&folder\autoexec.sas";

put '*name: autoexec.sas;'

/ "title1 '??? new project ???';"

/ "filename project '.';"

/ "libname library '.';";

file "&folder\initstmt.sas";

put '*name: initstmt.sas;'

/ '%put echo %sysfunc(getoption(initstmt));';

file "&folder\termstmt.sas";

put '*name: termstmt.sas;'

/ '%put echo %sysfunc(getoption(termstmt,keyexpand));';

stop;

run;

File sas.cmd centralizes the call to sas.exe.sas.cmd
rem name: sas.cmd

"C:\...\SASHome\SASFoundation\9.?\sas.exe" %*

Percent asterisk (%*) is the DOS syntax fornotes:
’pass all command-line parameters to program’.
DOS strings are numbered: %1 ... %n ; asterisk means ’all’.

sasv9.cfg File sasv9.cfg replicates the syntax of the main configuration file by pro-
viding an environment variable site inc which is then used in options
initstmt and termstmt.

Program 4 config test file: sasv9.cfg
/*name: sasv9.cfg */

-sasinitialfolder '.'

-set site_inc 'C:\SAS-projects\SAS-site\sas-includes'

-set site_inc '.'

-initstmt "%include '!site_inc\initstmt.sas';"

-termstmt "%include '!site_inc\termstmt.sas';"

Option sasinitialfolder is the argument to Display Manager command:notes:
File Save As; dot means ’here’: current folder.
Environment variable site inc has two values as a reminder that you may,
in the future, wish to provide access to subroutine files initstmt.sas and
termstmt.sas for all projects in a central folder.
The arguments of both contain a reference to the evar !site inc;
macro variables are referenced with an ampersand (&);
evars are referenced with a bang: exclamation point (!).

autoexec.sas
*name: autoexec.sas;

title1 '??? new project ???';

filename project '.';

libname library '.';

Options mautosource and sasautos are the pair of options of the autocallnotes:
of macros.

initstmt.sas Files initstmt.sas and termstmt.sas
*name: initstmt.sas;

%put echo %sysfunc(getoption(initstmt));

termstmt.sas *name: termstmt.sas;

%put echo %sysfunc(getoption(termstmt,keyexpand));

Function getoption has a required option of option-name and an optionalnotes:
argument of keyexpand which is used to format the value; both options
contain an evar reference: !site inc, which will be resolved when echoed.

Unit test: sasv9.cfg
This is the overview, which consists of a list of topics in this section.overview

• 01-config-test.*
• sections in 01-config-test
• 01-config-test.log, pg. 1, 2, 3, 4, 12

01-config-test.* Files 01-config-test.bat and 01-config-test.sas are used for a unit
test of the configuration file sasv9.cfg.

Program 5 01-config-test.*

rem name: 01-config-test.bat

sas 01-config-test -verbose -pagesize max

*name: 01-config-test.sas;

Option -verbose writes the settings of the system options to the log.notes:
Option -pagesize max removes page breaks from the log,
which may be 500+ lines; at 50 lines/page, more than 10 pages.

sections in
01-config-test

• Options specified in the config files:

• Options set internally at initialization:

• Options specified on the command line:

• Options set internally at initialization:

• Options set internally during session startup:

• Options set from locale:

• Options set internally at initialization:

01-config-test.log,
pg. 1

23 Options specified in the config file
24 C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg:
25 ---------------------------------------------------------------
26 CONFIG = "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"
27 ---------------------------------------------------------------

line 26, primary config file:notes:
compare to listing from program echo-config-main.sas, pg. 4;

28 Options specified in the config file
29 C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg:
30 ---------------------------------------------------------------
31 ...
32 SET = SASROOT "C:\program-files\SASHome\SASFoundation\9.4"
33 JREOPTIONS = (-DPFS_TEMPLATE=!SASROOT\tkjava\sasmisc\qrpfstpt.xml

. . .
46 SET = SASAUTOS ("!SASROOT\core\sasmacro"
47 ...
48 "!SASROOT\graph\sasmacro"
49 "!SASROOT\stat\sasmacro"
50 )

Lines 32–33 show the config file allocating an environment variable,notes:
sasroot and using it in an option, jreoptions;
lines 46–50 show its use in another environment variable, sasautos,
which is the the list of folders for the fileref and option sasautos.
Note the list of folders is enclosed in parentheses.

01-config-test.log,
pg 2 88 SET = MYSASFILES "?FOLDERID_Documents\My SAS Files\9.4"

89 SASUSER = "?FOLDERID_Documents\My SAS Files\9.4"
90 WORK = "!TEMP\SAS Temporary Files"
91 ...
92 SET = SASCFG "C:\program-files\SASHome\SASFoundation\9.4\nls\en"

?FOLDERID Documents line 88–89:notes:
question mark (?) is special character used to reference an API

!TEMP line 92, is a reference to a Windows environment variable
In previous versions, my-SAS-files SAS-user values fetched value of! →
CSIDL*
. . .01-config-test.log,

pg 3 111 ---------------------------------------------------------------
112 Options specified in the config file
113 C:\SAS-projects\2021-book\sas-startup-config-test-suite\sasv9.cfg:
114 ---------------------------------------------------------------
115 SASINITIALFOLDER = '.'
116 SET = site_inc 'C:\SAS-projects\SAS-site\sas-includes'
117 SET = site_inc '.'
118 INITSTMT = "%include '!site_inc\initstmt.sas';"
119 TERMSTMT = "%include '!site_inc\termstmt.sas';"
120 ---------------------------------------------------------------
121 Options set internally at initialization:
122 ---------------------------------------------------------------
123 CONFIG = ( "C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg"
124 "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"
125 "C:\SAS-projects\2021-book\sas-startup-config-test-suite\sasv9.cfg" )
126 ---------------------------------------------------------------
127 Options specified on the command line:
128 ---------------------------------------------------------------
129 SYSIN = 01-config-test
130 VERBOSE
131 PAGESIZE = max
132 ---------------------------------------------------------------
133 Options set internally at initialization:
134 ---------------------------------------------------------------
135 AUTOEXEC = "C:\SAS-projects\2021-book\sas-startup-config-test-suite\autoexec.sas"

Options set internally at initialization:notes:
the list of configuration files and order in which they are 'loaded',123–125:
note the use of the SAS HelpDesk verb loaded :! →
this means the last evar allocation or option assignment is the value used

Options specified on the command line:

command-line assignments 'loaded' after configuration file assignments129–131:
Options set internally at initialization:

option autoexec is set to value of sasinitialfolder135:
plus default filename: autoexec.sas

01-config-test.log,
pg. 4

155 NOTE: Copyright (c) 2002-2012 by SAS Institute Inc., Cary, NC, USA.

156 NOTE: SAS (r) Proprietary Software 9.?

157 ...

158 NOTE: SAS initialization used:

159 real time 2.55 seconds

160

161 ==== Processed Configuration File(s) ====

162 C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg

163 C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg

164 C:\SAS-projects\2021-book\sas-startup-config-test-suite\sasv9.cfg

165

166 ==== Environment Variable Options ====

167 <none>

168

169 Option Value

170 ====== =====

171 AUTOEXEC C:\SAS-projects\2021-book\sas-startup-config-test-suite\autoexec.sas

172 ...

beginning of notes from option sysin=01-config-test.sasnotes:
Processed Configuration File(s)

compare to page 3, lines 123–125162–164:

01-config-test.log,
pg. 12

550 NOTE: AUTOEXEC processing beginning; file is

551 C:\SAS-projects\2021-book\sas-startup-config-test-suite\autoexec.sas.

552

553 NOTE: Libref LIBRARY was successfully assigned as follows:

554 Engine: V9

555 Physical Name: C:\SAS-projects\2021-book\sas-startup-config-test-suite

556

557 NOTE: AUTOEXEC processing completed.

558

559 echo %include '!site_inc\initstmt.sas'

560 1 *name: 01-config-test.sas;

561

562 echo TERMSTMT=%include '.\termstmt.sas'

563 NOTE: SAS Institute Inc., SAS Campus Drive, Cary, NC USA 27513-2414

564 NOTE: The SAS System used:

565 real time 2.60 seconds

566 cpu time 0.46 seconds

notes:
option autoexec processing is550–557
%include 'autoexec.sas'\nosource2;
there is no echo of the filename statement
but any libref(s) allocated are duly noted
compare to termstmt, line 562559
line 1 from sysin560
compare to initstmt, line 559: keyexpand produces TERMSTMT=562

The next section shows polishing of echo from options initstmt and
termstmt.

Template for initstmt and termstmt

This section explains how to the use the start-up only options initstmt,overview
which statements are executed after the autoexec, and termstmt, which
statements are executed at endsas. The task here is to replicate the
beginning ... completed notes like those in autoexec processing.
This is the list of topics in this section.

• autoexec notes
• syntax
• initstmt.sas and log
• termstmt.sas and log

autoexec notes NOTE: AUTOEXEC processing beginning; file is
C:\SAS-projects\2021-book\sas-startup-config\autoexec.sas.

...
NOTE: AUTOEXEC processing completed.

Task 1 is how to write a note with a semicolon in it.syntax
semicolon This is accomplished by enclosing the semicolon and CR/LF in the %str()

function:
%str(;

)

Task 2 is parsing the value of the options, which contains two words.
The initstmt value is '%include "file-specification";'%scan
The %scan(string,n,delimiters) function is used to pick the 2nd word;
the %sysfunc(getoption(option-name) pair of functions is used to fetch
the value of the option.

Task 3 is to write a set of notes containing job information:
startup date+time-stamp, userid, and job-name;job info
this information is necessary to match the log with the program when pars-
ing logs of program runs.

Task 4: The read-only macro variable sysprocessid

is a 32-character hexadecimal string, the first 16 characters of which con-time
tain the program startup date+time-stamp.

initstmt.sas * name initstmt.sas;

%put Note initstmt processing beginning%str(;

)file is %scan(%sysfunc(getoption(initstmt)),2,%str( ));

%let _string = JOB-INFO:;

%put &_string datetime_hex16=%substr(&sysprocessid,1,16);

%put &_string datetime=%sysfunc

(putn(%substr(&sysprocessid,1,16)x,datetime21.2));

%put &_string &=sysuserid;

%put &_string %sysfunc(getoption(sysin,keyword)) #;

%symdel _string;

%put Note initstmt processing completed.;

log Note: initstmt processing beginning; file is '!site_inc\initstmt.sas'.

JOB-INFO: datetime_hex16=40D5944000000000

JOB-INFO: datetime=31JUN2021:11:23:58.13

JOB-INFO: SYSUSERID=Ronald

JOB-INFO: SYSIN=C:\SAS-projects\...\02-autoexec-test.sas #

Note: initstmt processing completed

termstmt.sas * name: termstmt.sas;

%put Note: termstmt processing beginning%str(;

)file is %scan(%sysfunc(getoption(termstmt,keyexpand)),2,%str( ));

%put echo list global _user_ mvars, if any;

%put _user_;

*include '!site_inc\echo-mvars.sas';

%put Note: termstmt processing completed.;

log Note: termstmt processing beginning; file is '.\termstmt.sas'.

echo list global _user_ mvars, if any

Note: termstmt processing completed

13



Unit test: autoexec.sas
This section shows the programs and log.overview

• 02-autoexec-test.*
• 02-autoexec-test.log, initstmt
• 02-autoexec-test.log, program
• 02-autoexec-test.log, termstmt

02-autoexec-test.* rem name: 02-autoexec-test.bat

sas 02-autoexec-test -echoauto -source -source2

*name: 02-autoexec-test.sas;

options pagesize=max;

filename _all_ list;

libname _all_ list;

%put echo mautosource=%sysfunc(getoption(mautosource));

%put %sysfunc(getoption(sasautos,keyword));

%let mvar2=text.2 for show sort order;

%let mvar1=text.1 for termstmt;

Options -echoauto and -source turn on the display of statements in thenotes:
autoexec file.
Option -source2 turns on the display of statements of %include files;
see files initstmt.sas and termstmt.sas.

02-autoexec-test.log,
initstmt NOTE: %INCLUDE (level 1) file !site_inc\initstmt.sas is file

C:\SAS-projects\2021-book\sas-startup-config\initstmt.sas.
1 +* name: initstmt.sas;
2 +%put Note: initstmt processing beginning%str(;
3 +)file is %scan(%sysfunc(getoption(initstmt)),2,%str( ));
Note: initstmt processing beginning; file is '!site_inc\initstmt.sas'
4 +
5 +%let _string = JOB-INFO;
6 +%put &_string: datetime_hex16=%substr(&sysprocessid,1,16);
JOB-INFO: datetime_hex16=41DCF8DA18B73B64
7 +%put &_string: datetime=%sysfunc
8 + (putn(%substr(&sysprocessid,1,16)x,datetime21.2));
JOB-INFO: datetime=11AUG2021:06:40:34.00
9 +%put &_string: &=sysuserid;
JOB-INFO: SYSUSERID=Ronald
10 +%put &_string: %sysfunc(getoption(sysin,keyword)) #;
JOB-INFO: SYSIN=C:\SAS-projects\2021-book\sas-startup-config\02-autoexec-test.sas #
11 +%symdel _string;
12 +
13 +%put Note: initstmt processing completed.;
Note: initstmt processing completed.
NOTE: %INCLUDE (level 1) ending.

02-autoexec-test.log,
program

1 *name: 02-autoexec-test.sas;
2 options pagesize=max;
3 filename _all_ list;
NOTE: Fileref= PROJECT

Physical Name= C:\SAS-projects\2021-book\sas-startup-configuration
4 libname _all_ list;
NOTE: Libref= LIBRARY

Scope= Program 02-autoexec-test
Engine= V9
Physical Name= C:\SAS-projects\2021-book\sas-startup-configuration
Filename= C:\SAS-projects\2021-book\sas-startup-configuration

...
5 %put echo mautosource=%sysfunc(getoption(mautosource));
echo mautosource=MAUTOSOURCE
6 %put %sysfunc(getoption(sasautos,keyword));
SASAUTOS=SASAUTOS
7 %let mvar2=text.2 for show sort order;
8 %let mvar1=text.1 for termstmt;

02-autoexec-test.log,
termstmt

NOTE: %INCLUDE (level 1) file !site_inc\termstmt.sas is file
C:\SAS-projects\2021-book\sas-startup-configuration\termstmt.sas.

2 +* name: termstmt.sas;
3 +%put Note: termstmt processing beginning%str(;
4 +)file is %scan(%sysfunc(getoption(termstmt,keyexpand)),2,%str( ));
Note: termstmt processing beginning; file is '.\termstmt.sas'
5 +
6 +%put echo list global _user_ mvars, if any;
echo list global _user_ mvars, if any
7 +%put _user_;
GLOBAL MVAR1 text.1 for termstmt
GLOBAL MVAR2 text.2 for show sort order
8 +*include '!site_inc\echo-mvars.sas';
9 +
10 +%put Note: termstmt processing completed.;
Note: termstmt processing completed.
NOTE: %INCLUDE (level 1) ending.

Summary
api: Wikipedia editors, et al. [9]Suggested Reading
csidl: Microsoft Corp. [5]; known-folder-id: Microsoft Corp. [6]Microsoft Corp:
csidl: [7]; files used by SAS: SAS Institute [8]SAS Institute:
Fehd [3] autoexec companioncompanions:
Fehd [4] sysparm companion
Fehd [1] writing testing-aware programs using command-line optionstesting:
echoauto and verbose

Fehd [2] using sql for list processing;sql
see program ProcSQL-D-Macros-select-into-list-ordered.sas for use
in termstmt.sas

Configuration files contain two kinds of statements: allocations of environ-Conclusion
ment variables and options’ assignments. SAS configuration files use envi-
ronment variables as arguments for options. Environment variables contain
list of folders, libraries, with files that contain its functions, procedures, and
macros. Writing configuration files can be a simple matter. The purpose
of this paper is to provide a suite of tools that make testing user-written
configuration files easier.

